Computations for the Introduction

The work of Kevin McCurley

The paper

Explicit Estimates for Primes in Progressions, by Kevin S. McCurley

contains the following as Theorem 2.1 on page 267.

THEOREM 2.1. Let $T \geq 1$ and χ be a primitive nonprincipal character modulo q. If $0 < \eta \leq \frac{1}{2}$, then

$$\left| N(T,\chi) - \frac{T}{\pi}\log\frac{qT}{2\pi e}\right| < C_1 \log qT + C_2,$$

where

$$C_1 = \frac{1+2\eta}{\pi\log 2},$$

$$C_2 = 0.3058 - 0.268\,\eta + \frac{4\log\zeta(1+\eta)}{\log 2} - \frac{2\log\zeta(2+2\eta)}{\log 2} + \frac{2}{\pi}\log\zeta\left(\frac{3}{2}+2\eta\right).$$

For example, with $\eta = \frac{1}{2}$, we have $C_1 \leq 0.919$ and $C_2 \leq 5.340$, and with $\eta = 0.05$, we have $C_1 \leq 0.506$ and $C_2 \leq 16.989$.

First, we note that the minimum value of C_1 is $\frac{1}{\pi\,\text{Log}[2]} > 0.45$.

```
In[1]:= N[1/(π Log[2]), 10]

Out[1]= 0.4592240943
```

This leads to the following upper and lower bounds on $N(T, \chi)$.

In[2]:=
```
upper["McCurley", T_, q_] := Module[{C1, C2, eta},
   C1[eta_] = (1 + 2 eta)/(Pi * Log[2]);
   C2[eta_] = 0.3058 - 0.268 eta +
     4 Log[Zeta[1 + eta]]/Log[2] - 2 Log[Zeta[2 + 2 eta]]/Log[2] + 2/π Log[Zeta[3/2 + 2 eta]];
   Floor[T/π Log[q T/(2 π E)]] + First[NMinimize[
       {C1[eta] Log[q T] + C2[eta], 0 < eta ≤ 1/2}, {eta}]]]];
lower["McCurley", T_, q_] := Module[{C1, C2, eta},
   C1[eta_] = (1 + 2 eta)/(Pi * Log[2]);
   C2[eta_] = 0.3058 - 0.268 eta +
     4 Log[Zeta[1 + eta]]/Log[2] - 2 Log[Zeta[2 + 2 eta]]/Log[2] + 2/π Log[Zeta[3/2 + 2 eta]];
   Ceiling[T/π Log[q T/(2 π E)]] - First[NMinimize[
       {C1[eta] Log[q T] + C2[eta], 0 < eta ≤ 1/2}, {eta}]]]];
```

For low heights, these bounds are very bad. In fact, at $T = 1$, the lower bound is negative!

In[4]:=
```
McCurleyRegion = Show[
  {Plot[Block[{eta = 1/2 (-1 + π C1 Log[2])}, 0.3058 - 0.268 eta + 4 Log[Zeta[1 + eta]]/Log[2] -
       2 Log[Zeta[2 + 2 eta]]/Log[2] + 2/π Log[Zeta[3/2 + 2 eta]]],
    {C1, 1/(π Log[2]), 2/(π Log[2])}, AspectRatio → 1/3, Filling → Top,
    FillingStyle → LightBlue, PlotRange → {{0, 0.9}, {-3.2, 20}}],
   Graphics[Text["McCurley", {0.6, 13}, {0, 0}, {2, -1}]]}]
```

Out[4]=



In[5]:=
```
Plot[{lower["McCurley", 1, Exp[LogqT]], upper["McCurley", 1, Exp[LogqT]]},
 {LogqT, 2, 27}, PlotLegends → {"lower", "upper"}, AxesLabel → {"log qT", "N(T,chi)"}]
```

Out[5]=


In[6]:=
```
(* NOTE: DirichletCharacters` is O'Bryant's package,
thoroughly implementing Conrey's notation for dirichlet characters,
available on GitHub. *)
<< DirichletCharacters`
Sum[NumberOfPrimitiveCharacters[q], {q, 1, 10^5}]
```

Out[7]= 1 847 865 075

How many zeros of height at most 1, and conductor at most 10^5?

Up to $q = 9082$, the optimal value of η is $1/2$.

In[8]:=
```
sum1 = Sum[NumberOfPrimitiveCharacters[q] * Floor[
   1/π Log[q/(2 π E)] + (1 + 2 (1/2))/(Pi * Log[2]) Log[q] + 3058/10000 - 268/1000 (1/2) + (4 Log[Zeta[1 + (1/2)]])/Log[2] -
    (2 Log[Zeta[2 + 2 (1/2)]])/Log[2] + 2/π Log[Zeta[3/2 + 2 (1/2)]]], {q, 1, 9082}]

starttime = AbsoluteTime[];
Monitor[sum2 = Sum[NumberOfPrimitiveCharacters[q] * Floor[1/π Log[q/(2 π E)] +
      First[FindMinimum[{(1 + 2 eta)/(Pi * Log[2]) Log[q] + 0.3058 - 0.268 eta + (4 Log[Zeta[1 + eta]])/Log[2] -
          (2 Log[Zeta[2 + 2 eta]])/Log[2] + 2/π Log[Zeta[3/2 + 2 eta]], 0 < eta ≤ 1/2},
        {eta}, PrecisionGoal → 6, WorkingPrecision → MachinePrecision]]],
   {q, 9083, 10^5}], {ProgressIndicator[q, {9083, 10^5}], q,
   ((((AbsoluteTime[] - starttime) (10^5 - 9082))/(q - 9082)) - (AbsoluteTime[] - starttime))/60
   min }]

sum1 + sum2
```

Out[8]= 222 849 037

Out[10]= 32 233 356 552

Out[11]= 32 456 205 589

The work of Tim Trudgian

The paper

An Improved Upper Bound for the Error in the Zero-Counting Formulae for Dirichlet L-functions and Dedekind Zeta-Functions, by T. S. Trudgian. Mathematics of Computation, 2014.

contains the following as Theorem 1.

THEOREM 1. Let $T \geq 1$ and χ be a primitive nonprincipal character modulo q. Then

$$\left| N(T, \chi) - \frac{T}{\pi} \log \frac{q\,T}{2\,\pi e} \right| < 0.315 \log q\,T + 6.455.$$

In addition, if the right side is written as $C_1 \log q\,T + C_2$, on may use the values of C_1 and C_2 contained in Table 1.

In[12]:=
```
c1c2pairs1 =
  {{.247, 9.359}, {.264, 8.049}, {.281, 7.323}, {.298, 6.828}, {.315, 6.455},
   {.332, 6.156}, {.349, 5.907}, {.365, 5.694}, {.382, 5.506}, {.399, 5.338}};
c1c2pairs10 = {{.247, 8.949}, {.264, 7.640}, {.281, 6.914},
   {.298, 6.419}, {.315, 6.046}, {.332, 5.747}, {.349, 5.498},
   {.365, 5.284}, {.382, 5.096}, {.399, 4.928}};
```

In[14]:=
```
(* IMPORTANT: These are not proved, as Trudgian has errors. *)
upper["Trudgian", T_, q_] :=
  Floor[T/π Log[q T/(2 π E)] + If[T ≥ 10, Min[Map[{Log[q T], 1}.# &, c1c2pairs10]],
     Min[Map[{Log[q T], 1}.# &, c1c2pairs1]]]];
lower["Trudgian", T_, q_] := Ceiling[T/π Log[q T/(2 π E)] -
    If[T ≥ 10, Min[Map[{Log[q T], 1}.# &, c1c2pairs10]],
     Min[Map[{Log[q T], 1}.# &, c1c2pairs1]]]];
```

In[16]:=
```
Sum[upper["Trudgian", 1, q] * NumberOfPrimitiveCharacters[q], {q, 10^5}]
```

Out[16]= 21 880 443 454

In[17]:=
```
Trudgianpic = Show[McCurleyRegion, Map[Graphics[{Black, Point[#]}] &,
   Union[c1c2pairs1, c1c2pairs10]], McCurleyRegion,
  Graphics[Text["Trudgian", {0.35, 7.5}, {0, 0}, {4, -1}]], ImageSize → 500]
```

Out[17]=



Corollaries of the Main Bound

In[18]:=
```
MainBound[L_] = 22737/100000 L + 2 Log[1 + L] - 1/2;
```

Corollary 1.2

We need to add $\frac{1}{4}$ to Main bound to compensate for removing b_χ from the left side. And we use

$$\mathrm{Log}[q\,T] == \mathrm{Log}\left[\frac{q\,(T+2)}{2\,\pi}\right] + \mathrm{Log}[2\,\pi] - \mathrm{Log}\left[1+\frac{2}{T}\right] == L + \mathrm{Log}[2\,\pi] - \mathrm{Log}\left[1+\frac{2}{T}\right]$$

In[19]:=
```
Simplify[Log[q T] - (Log[q (T + 2)/(2 π)] + Log[2 π] - Log[1 + 2/T]),
 Assumptions → T ≥ 5 / 7 && q ≥ 3]
```

Out[19]=
```
0
```

In[20]:=
```
f[{C1_, C2_}, L_, T_] = (C1 * (L + Log[2 π] - Log[1 + 2/T]) + C2) - (MainBound[L] + 1/4);
Reduce[D[f[{C1, C2}, L, T], L] == 0 && C1 ≥ 22737/100000 && L ≥ 1567/1000, L]
Reduce[D[f[{C1, C2}, L, T], T] ≥ 0 && T ≥ 5 / 7]
```

Out[21]=
$$\frac{22737}{100000} < C1 \le \frac{258365879}{256700000}\ \&\&\ L == \frac{222737 - 100000\,C1}{-22737 + 100000\,C1}$$

Out[22]=
$$C1 \ge 0\ \&\&\ T \ge \frac{5}{7}$$

So for a particular C_1, C_2, the bound is at its worst with $T = \frac{5}{7}$ and $L = \frac{222737-100000\,C1}{-22737+100000\,C1}$.

In[23]:=
```
N[Block[{C1 = 247/1000, C2 = 6894/1000}, f[{C1, C2}, (222737 - 100000 C1)/(-22737 + 100000 C1), 5/7]], 30]
N[Block[{C1 = 298/1000, C2 = 4358/1000}, f[{C1, C2}, (222737 - 100000 C1)/(-22737 + 100000 C1), 5/7]], 30]
N[Block[{C1 = 1/3, C2 = 11/3}, f[{C1, C2}, (222737 - 100000 C1)/(-22737 + 100000 C1), 5/7]], 30]
N[Block[{C1 = 1/4, C2 = 20/3}, f[{C1, C2}, (222737 - 100000 C1)/(-22737 + 100000 C1), 5/7]], 30]
```

Out[23]= 0.000893469412795004028709361754538

Out[24]= 0.000332095661244672839980977383777

Out[25]= 0.102709992116922784516314715332

Out[26]= 0.0565036667926453072329320844059

The bound $\text{Log}\left[\frac{q\,(T+2)}{2\,\pi}\right] \geq 1.567$ implies that $\text{Log}[q\,T] \geq 1.567 + \text{Log}[2\,\pi] - \text{Log}\left[1 + \frac{2}{T}\right]$, from is equivalent (as q and T are positive) to

$$T \geq 2\left(\frac{\pi}{q}\,e^{1.567} - 1\right).$$

Under the assumption that $q \geq 12$, this is weaker that $T \geq \frac{5}{7}$. Therefore, we need only to check by hand the L-functions with conductor 3, 4, 5, 7, 8, 9, and 11. And we only need to go up to the height $2\left(\frac{\pi}{q}\,e^{1.567} - 1\right)$. We go up to height $\frac{2\,\pi}{q}\,e^{6}$ because that's our data.

In[50]:=
```
(*PC[q] is a list of primitive characters with conductor q,
including 1 from  each conjugate pair *)
PC[q_] := PC[q] = Select[PrimitiveCharacters[q],
    ConreyIndex[#] ≤ ConreyIndex[ConjugateCharacter[#]] &];
```

In[51]:=
```
AbsoluteTiming[Monitor[Do[PC[q], {q, 1, 11}], q]]
```

Out[51]= {0.019399, Null}

In[52]:=
```
(* This output should be a list of nonnegative numbers *)
With[{C1 = 247/1000, C2 = 6894/1000},
 Table[
  q = Conductor[char];
  {maxheight, zeros} = LSeriesZeroHeights[char];
  zeros = Sort[Abs[zeros]];
  countofzeros = 0;
  extremes = Sort[Flatten[Table[
      countofzeros++;
      {C1 Log[q zeros[[k]]] + C2 - Abs[countofzeros - zeros[[k]]/π Log[q zeros[[k]]/(2 π ⅇ)]],
       C1 Log[q zeros[[k]]] + C2 - Abs[countofzeros - 1 - zeros[[k]]/π Log[q zeros[[k]]/(2 π ⅇ)]]}
      , {k, Length[zeros]}]]];
  Min[extremes],
  {char, Flatten[Table[PC[q], {q, 3, 11}]]}]]
```

Out[52]=
```
{6.0541005342, 6.1991541054, 6.84758047997, 6.08519230253,
 6.53789015385, 6.41759845134, 6.0043368817, 6.30957940728,
 6.3086713886, 6.74271565078, 6.63014637993, 6.73176134109,
 6.8002679953, 6.45263104152, 6.44672967320, 6.07871928408}
```

In[53]:=
```
(* This output should be a list of nonnegative numbers *)
With[{C1 = 298/1000, C2 = 4358/1000},
 Table[
  q = Conductor[char];
  {maxheight, zeros} = LSeriesZeroHeights[char];
  zeros = Sort[Abs[zeros]];
  countofzeros = 0;
  extremes = Sort[Flatten[Table[
      countofzeros++;
      {C1 Log[q zeros[[k]]] + C2 - Abs[countofzeros - zeros[[k]]/π Log[q zeros[[k]]/(2 π ⅇ)]],
       C1 Log[q zeros[[k]]] + C2 - Abs[countofzeros - 1 - zeros[[k]]/π Log[q zeros[[k]]/(2 π ⅇ)]]}
      , {k, Length[zeros]}]]];
  Min[extremes],
  {char, Flatten[Table[PC[q], {q, 3, 11}]]}]]
```

Out[53]=
```
{3.84523209197, 3.96473670608, 4.51113380643, 3.77278335197,
 4.24233382168, 4.09816907458, 3.75900576193, 3.94461962182,
 4.10849498186, 4.41722107371, 4.30918495669, 4.38262580179,
 4.48557290050, 4.13965087168, 4.04362929949, 3.71127641630}
```

In[54]:=
```
c1c2pairs1
c1c2pairs10
```

Out[54]=
```
{{0.247, 9.359}, {0.264, 8.049}, {0.281, 7.323}, {0.298, 6.828}, {0.315, 6.455},
 {0.332, 6.156}, {0.349, 5.907}, {0.365, 5.694}, {0.382, 5.506}, {0.399, 5.338}}
```

Out[55]=
```
{{0.247, 8.949}, {0.264, 7.64}, {0.281, 6.914}, {0.298, 6.419}, {0.315, 6.046},
 {0.332, 5.747}, {0.349, 5.498}, {0.365, 5.284}, {0.382, 5.096}, {0.399, 4.928}}
```

In[56]:= $\mathtt{C2func[C1\_] = -f\left[\{C1, 0\}, \frac{222737 - 100000\,C1}{-22737 + 100000\,C1}, \frac{5}{7}\right];}$

In[57]:= $\mathtt{Newpic = Show\left[Trudgianpic, Plot\left[C2func[C1], \left\{C1, \frac{22737}{100000}, 0.9\right\}, PlotRange \to \{\{0, 0.9\}, \{-3.2, 20\}\}, Filling \to Top\right], ListPlot\left[\left\{\left\{\frac{247}{1000}, \frac{6894}{1000}\right\}, \left\{\frac{298}{1000}, \frac{4358}{1000}\right\}\right\}\right]\right]}$

Out[57]=


In[58]:=
```
Export[FileNameJoin[{NotebookDirectory[], "Before.pdf"}], Newpic];
```

In[59]:=
```
<< DirichletCharacters`
```

In[60]:=
```
a0chars = Sum[NumberOfPrimitiveCharacters[q, 0] * 0, {q, 1, 36}] +
   Sum[NumberOfPrimitiveCharacters[q, 0] * 1, {q, 37, 148}] +
   Sum[NumberOfPrimitiveCharacters[q, 0] * 2, {q, 149, 844}] +
   Sum[NumberOfPrimitiveCharacters[q, 0] * 5, {q, 845, 1616}] +
   Sum[NumberOfPrimitiveCharacters[q, 0] * 6, {q, 1617, 6256}] +
   Sum[NumberOfPrimitiveCharacters[q, 0] * 7, {q, 6257, 25 252}] +
   Sum[NumberOfPrimitiveCharacters[q, 0] * 8, {q, 25 253, 10^5}];
a1chars = Sum[NumberOfPrimitiveCharacters[q, 1] * 0, {q, 1, 12}] +
   Sum[NumberOfPrimitiveCharacters[q, 1] * 1, {q, 13, 42}] +
   Sum[NumberOfPrimitiveCharacters[q, 1] * 2, {q, 43, 408}] +
   Sum[NumberOfPrimitiveCharacters[q, 1] * 3, {q, 409, 844}] +
   Sum[NumberOfPrimitiveCharacters[q, 1] * 5, {q, 845, 905}] +
   Sum[NumberOfPrimitiveCharacters[q, 1] * 6, {q, 906, 3425}] +
   Sum[NumberOfPrimitiveCharacters[q, 1] * 7, {q, 3426, 55 727}] +
   Sum[NumberOfPrimitiveCharacters[q, 1] * 8, {q, 55 728, 10^5}];
a0chars +
 a1chars
```

Out[62]= 14 431 705 483

Main Theorem Bound

From the Main Theorem, N(1,chi) =0 if $\mathrm{Log}\left[\frac{q(3)}{2\pi}\right] \leq 1.567$, whence N(1,chi)=0 for $q \leq 10$.

In[63]:=
```
Sum[NumberOfPrimitiveCharacters[q] *
  Floor[1/π Log[q/(2 π E)] + 1/4 + 0.22737 Log[q (3)/(2 π)] + 2 Log[1 + Log[q (3)/(2 π)]] - 1/2], {q, 11, 10^5}]
```

Out[63]= 16 461 465 486

Lower Bound Computations

In[64]:=
```
Clear[q]
```

In[65]:=
```
b[a_] = If[a == 0, -1 / 4, 1 / 4];
lower["New", T_, q_, a_] = T/π Log[q T/(2 π E)] + b[a] - MainBound[Log[q (T + 2)/(2 π)]]
```

Out[66]= $\frac{1}{2} + \mathrm{If}\left[a == 0, -\frac{1}{4}, \frac{1}{4}\right] + \frac{T\,\mathrm{Log}\left[\frac{q\,T}{2\,e\,\pi}\right]}{\pi} - \frac{22\,737\,\mathrm{Log}\left[\frac{q\,(2+T)}{2\,\pi}\right]}{100\,000} - 2\,\mathrm{Log}\left[1 + \mathrm{Log}\left[\frac{q\,(2+T)}{2\,\pi}\right]\right]$

In[67]:=
```
N[lower["New", 1, 1.3 * 10^47, 0], 10]
N[lower["New", 1, 1.3 * 10^47, 1], 10]
```

Out[67]= 0.00226851

Out[68]= 0.502269

In[69]:=
```
N[lower["New", 2, 1.2 * 10^7, 0], 10]
N[lower["New", 2, 1.2 * 10^7, 1], 10]
```

Out[69]= 0.00967797

Out[70]= 0.509678